Modeling Vacuum Tubes

Part II

In the February 1994 issue of the *Intusoft Newsletter*, we introduced a very accurate triode vacuum tube model that makes use of the IsSpice4 behavioral modeling capability (in-line equations). In part II of this article we will extend the development to include models for pentodes or tetrodes.

A Realistic Pentode Model

Pentodes users have had more problems then triode enthusiasts when trying to simulate their favorite component, since no realistic models have existed up to now. In references [1,2] equations are given for the plate-to-cathode current flow as:

$$I_K = K_T\left(\frac{E_B}{\mu} + E_C\right)^{\frac{3}{2}} \text{ ; triodes } \quad I_K = K_T\left(\frac{E_{C2}}{\mu} + E_C\right)^{\frac{3}{2}} \frac{2}{\pi}\arctan\left(\frac{E_B}{10}\right) \text{ ; pentodes}$$

While the triode equation can be solved with the SPICE 2 style polynomial [1], both equations are more easily simulated using the in-line equation feature of IsSpice4. For example:

```
B1 1 2 I = KT * (EC2/μ + EC)^1.5 * 2/pi * ATAN(EB/10)
```

where KT and μ are constants and EC2, EC, and EB are node names. Unfortunately, these equation are only approximations lacking sufficient accuracy with respect to positive grid voltage, grid current, and high plate currents for many applications.

The model for the more realistic pentode in Table 1 was derived from the triode model in part I, but takes into account the specific properties attributable to several grids and their interactions. The model takes into account the law for the G2 (second grid) current, the effects of the G2 voltage on the anode current and the G1 current (in the case of positive G1 voltage). Even the effects of the virtual cathode (quick reduction of the cathode current when Va goes below 40 Volts) have been modeled. The pentode subcircuit is made up of interelectrode capacitances, a heater subcircuit which can be excluded, and the main PENT1 subcircuit. The PENT1 portion is parameterized, allowing virtually any tube to be modeled. The parameters are defined in Table 2.

Figures 7 and 8 show the results for a sinusoidal input (8V 2kHz) and the characteristic output curves (IK versus VA, for several VG1 voltages), which are obtained with the new pentode model for a typical set of pentode parameters.

Table 1, Forward and reverse conditions are treated in the pentode model, as well as saturation. Model parameters for a typical tube are shown below.

```
.SUBCKT ELXXXX  1   2     3     4     5  6
*                  Anode Grid2 Grid1 Cathode F   F' COPYRIGHT EXCEM, 1993
X1 1 2 3 4 10 PENT1 {SFS=0.7 VBIG=-0.9 VBIA=-1.3 MUG2=17 MUA=15000 RMU=0.5
+ VMU=-20 SFMU=1.6 K=5.4E-3 RK=0.08 VK=-20 SFK=1.6 SIGMA1=0.05
+ ALPHA1=5.2 SFG1=3.5 SIGMA2=0.12 ALPHA2=0.06 SFG2=2.3 VCCR=0.58 SFVC=0.33}
X2 5 6 10 HEAT1 {INOM=0.15 VNOM=6.3 LAMBDA=1 RCOOL=3 TCTE=10 TNOM=1150 INIT=100
+ W=2.045 ISAT=0.690}  ; Heater can be replaced with 2 statements "RF 5 6 42" & "VH 10 0 99m".
C2 1 2 1.5P
C3 3 1 0.5P
C4 2 3 1.6P
C5 3 4 4P
C6 3 5 4P
.ENDS
******************
.SUBCKT PENT1   A  G2  G1  C  ISAT   ; COPYRIGHT EXCEM, 1993
B1 15 0 V = V(G1) - V(C) < -1U ?  {K} * (1 + {RK} *
+ ((V(G1) - V(C)) / {VK})^{SFK}) / (1 + ((V(G1) - V(C)) / {VK})^{SFK}) : {K}
B2 16 0 V = V(G1) - V(C) < -1U ?  (1 + {RMU} * ((V(G1) - V(C)) / {VMU})^{SFMU}) /
+ (1 + ((V(G1) - V(C)) / {VMU})^{SFMU}) : 1
E1 17 0 16 0 {MUG2}
E2 18 0 16 0 {MUA}
B4 9 0 V = V(G1) - V(C) - {VBIG} + (V(A) - V(C) - {VBIA}) / (V(18) + 1U)+(V(G2) - V(C)) / (V(17)+1U)
B6 10 0 V = V(9) > 1P ? V(15) * V(9)^1.5 / (V(ISAT) + 1P) : 0
B7 12 0 V = V(10) < {SFS} ? V(10) * (V(ISAT) + 1P) :
+ (V(ISAT) + 1P) * ({SFS} + (V(10) - {SFS}) * {1-SFS} / ({1 - 2 * SFS} + V(10)))
B8 14 0 V = V(G2) - V(C) + {MUG2 / MUA} * (V(A) - V(C)) > 0.1M ?
+ V(G2) - V(C) + {MUG2 / MUA} * (V(A) - V(C)) / {ALPHA1} : 0.2M
B9 28 0 V = V(G1) - V(C) > {VBIG + 10U} ? V(14) > 0.1M ? ((V(G1) - V(C) - {VBIG} +
+ {SIGMA1^(1/SFG1)} * V(14)) / (V(G1) - V(C) - {VBIG} + V(14)))^{SFG1} : 0
B10 8 0 V = V(G1) - V(C) < 0 ? V(28) * (({VBIG+10U} + V(C) - V(G1)) / {VBIG+10U}) : V(28)
B11 21 0 V = V(A) - V(C) > {VBIA + 0.1M} ? (V(A) - V(C) - {VBIA}) / {ALPHA2} : 0.2M
B12 32 0 V = V(G2) - V(C) > {VBIG+10U} ? V(21) > 0.1M ?
+((V(G2) - V(C) - {VBIG} + {SIGMA2^(1/SFG2)} * V(21)) / (V(G2) - V(C) - {VBIG} + V(21)))^{SFG2} : 0
B13 22 0 V = V(G2) - V(C) < 0 ? V(32) * (({VBIG+10U} + V(C) - V(G2)) / {VBIG+10U}) : V(32)
B14 23 0 V = V(22) - {SIGMA2} > 1P ? V(12) * (1-{VCCR} * (V(22) - {SIGMA2})^{SFVC}) : V(12)
B15 G1 C I = V(8) * V(23)
R15 G1 C 100MEG
B16 G2 C I = (1 - V(8)) * V(22) * V(23)
R16 G2 C 100MEG
B17 A C I = (1 - V(8)) * (1 - V(22)) * V(23)
R17 A C 100MEG
.ENDS
```

See Table 2 for notes on the B elements and parameters. See Feb. '94 issue for information on the heater model.



Figure 7, Cross-probed waveforms show the pentode tube response to sinusoidal input (8V, 2kHz) at the cathode.

Table 2, The generic parameters for the pentode subcircuit (PENT1) reveal the tremendous versatility of the model.

The Triode Parameters Are:

SFS	Shape factor of the saturation law.
VBIG	Contact potential of the grid G1 (above this value grid current may start to flow).
VBIA	Contact potential of the anode.
MUG2	Amplification factor for G2 at slightly negative G1 voltage.
MUA	Amplification factor for A at slightly negative G1 voltage.
RMU	Reduction factor for MU at very negative G1 voltage.
MU	Grid voltage for mid-range MU (negative).
SFMU	Shape factor for MU reduction law.
K	Perveance at slightly negative G1 voltage.
RK	Perveance reduction factor at very negative G1 voltage.
VK	Grid voltage for mid-range perveance (negative).
SFK	Shape factor for perveance reduction law.
SIGMA1	Effective cross-section of G1 relative to the anode and G2.
ALPHA1	Grid G1 current amplification factor.
SFG1	Shape factor of the grid G1 current law.
SIGMA2	Effective cross-section of G2 relative to the anode.
ALPHA2	Grid G2 current amplification factor.
SFG2	Shape factor of the grid G2 current law.
VCCR	Virtual cathode current ratio.
SFVC	Shape factor of the virtual cathode current law.

Model Notes: B7 contains an arbitrary saturation law modeled by the shape factor SFS to match the available (?) data. SFS should be between 0 and 1, and the lower SFS, the sloppier the saturation law. V(15) is the effective perveance. V(16) is the factor used to establish both effective MU coefficients. V(17) is the effective MUG2 and V(18) is the effective MUA. (B14, V(23)) When the virtual cathode is present, this factor describes the decrease in cathode current (see Terman p. 192). The model describes only the static behavior of the pentode, and neglects secondary emission. It is assumed that G2 is always very positive with respect to the cathode. The heater parameters are described in part 1 of this article (Feb. 1994 issue).

Figure 9 shows the use of a pentode in an cascode amplifier circuit similar to the one which appeared in part I of this article and used a dual triode combination.



Figure 8, DC characteristics for a typical pentode. The graph clearly shows the nonlinear effects of anode and grid current versus anode and grid G1 voltage.



Figure 9, A cascode amplifier using the new pentode model.

The response of this amplifier to a sinusoid of 100 kHz with an amplitude of 1.6 Volts pk-pk gives an output amplitude of about 430 Volts pk-pk. As in the case of the triode version, there is a slight distortion at this level, however with different characteristics. VGMOS is the signal at the input and VA is the signal at the output of the pentode.

Waveforms for the pentode's cathode (VK) and grid VG2 are also shown. The non-linear shape of the cathode's voltage is very evident. The strong feedback that creates this particular waveshape should be given credit for the almost linear response obtained at the output. The curve of the Figure 10 gives the frequency response of the pentode amplifier.

In order to verify the models an amplifier similar to Figure 9 was constructed and simulated with good correlation. The amplifier used a 2kV beam tetrode and 1kV-1.2kV power supply to deliver a peak-to-peak output signal of about 960 Volts. This amplifier was designed for the measurement of the transfer



Figure 10, Frequency response of the pentode amplifier.

admittance of shielded cable in the 100 kHz to 10 MHz frequency range. With IsSpice and these new vacuum tube models audio engineers can evaluate their designs more accurately and realistically then ever before. Those tube enthusiasts who are interested in vacuum tube SPICE models for specific parts should contact Intusoft directly.

References

[1] , C. Hymowitz, L. Meares, "SPICE Applications Handbook 2nd Edition", Intusoft, March 1994

[2] Scott Reynolds, "Vacuum Tube Models For SPICE Simulations", Glass Audio, April 1993

[3] Frederick E. Terman, "Electronic and Radio Engineering", McGraw-Hill, 1955

Tube models were created by: EXCEM 12, Chemin des Hauts de Clairefontaine 78580 MAULE FRANCE Tel 33 (1) 34 75 13 65, Fax 33 (1) 34 75 13 66